Rules for integrands of the form $(a + b\,\text{Sech}[c + d\,x^n])^p$

1: $\int (a + b\,\text{Sech}[c + d\,x^n])^p\,dx$ when $\frac{1}{n} \in \mathbb{Z}^+ \land p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $-1 \le n \le 1 \land n \ne 0$, then $F[x^n] = \frac{1}{n}\,\text{Subst}\left[x^{\frac{1}{n}-1} F[x],\, x,\, x^n\right] \partial_x x^n$

Note: If $\frac{1}{n} \in \mathbb{Z}^-$, resulting integrand is not integrable.

Rule: If $\frac{1}{n} \in \mathbb{Z}^+ \land p \in \mathbb{Z}$, then

$$\int (a + b\,\text{Sech}[c + d\,x^n])^p\,dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\int x^{\frac{1}{n}-1} (a + b\,\text{Sech}[c + d\,x])^p\,dx,\, x,\, x^n\right]$$

Program code:

```
Int[(a_.+b_.*Sech[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(1/n-1)*(a+b*Sech[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,p},x] && IGtQ[1/n,0] && IntegerQ[p]
```

```
Int[(a_.+b_.*Csch[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(1/n-1)*(a+b*Csch[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,p},x] && IGtQ[1/n,0] && IntegerQ[p]
```

X: $\int (a + b \operatorname{Sech}[c + d x^n])^p \, dx$

Rule:

$$\int (a + b \operatorname{Sech}[c + d x^n])^p \, dx \rightarrow \int (a + b \operatorname{Sech}[c + d x^n])^p \, dx$$

Program code:

```
Int[(a_.+b_.*Sech[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[(a+b*Sech[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x]
```

```
Int[(a_.+b_.*Csch[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[(a+b*Csch[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x]
```

S: $\int (a + b \operatorname{Sech}[c + d u^n])^p \, dx$ when $u == e + f x$

Derivation: Integration by substitution

Rule: If $u == e + f x$, then

$$\int (a + b \operatorname{Sech}[c + d u^n])^p \, dx \rightarrow \frac{1}{f} \operatorname{Subst}\left[\int (a + b \operatorname{Sech}[c + d x^n])^p \, dx, x, u\right]$$

Program code:

```
Int[(a_.+b_.*Sech[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*Sech[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```

```
Int[(a_.+b_.*Csch[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*Csch[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```

N: $\int (a + b\,\mathrm{Sech}[u])^p\,dx$ when $u == c + d\,x^n$

Derivation: Algebraic normalization

Rule: If $u == c + d\,x^n$, then

$$\int (a + b\,\mathrm{Sech}[u])^p\,dx \longrightarrow \int \left(a + b\,\mathrm{Sech}\left[c + d\,x^n\right]\right)^p\,dx$$

Program code:

```
Int[(a_.+b_.*Sech[u_])^p_.,x_Symbol] :=
  Int[(a+b*Sech[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

```
Int[(a_.+b_.*Csch[u_])^p_.,x_Symbol] :=
  Int[(a+b*Csch[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

Rules for integrands of the form $(e\,x)^m\,(a+b\,\text{Sech}[c+d\,x^n])^p$

1. $\int x^m\,(a+b\,\text{Sech}[c+d\,x^n])^p\,dx$

1: $\int x^m\,(a+b\,\text{Sech}[c+d\,x^n])^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}^+\land p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m\,F[x^n]=\frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\partial_x x^n$

Note: If $\frac{m+1}{n}\in\mathbb{Z}^-$, resulting integrand is not integrable.

Rule: If $\frac{m+1}{n}\in\mathbb{Z}^+\land p\in\mathbb{Z}$, then

$$\int x^m\,(a+b\,\text{Sech}[c+d\,x^n])^p\,dx\;\longrightarrow\;\frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\,(a+b\,\text{Sech}[c+d\,x])^p\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Sech[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Sech[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IGtQ[Simplify[(m+1)/n],0] && IntegerQ[p]
```

```
Int[x_^m_.*(a_.+b_.*Csch[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Csch[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IGtQ[Simplify[(m+1)/n],0] && IntegerQ[p]
```

X: $\int x^m \left(a+b\,\text{Sech}\left[c+d\,x^n\right]\right)^p dx$

Rule:

$$\int x^m \left(a+b\,\text{Sech}\left[c+d\,x^n\right]\right)^p dx \longrightarrow \int x^m \left(a+b\,\text{Sech}\left[c+d\,x^n\right]\right)^p dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Sech[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[x^m*(a+b*Sech[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x]
```

```
Int[x_^m_.*(a_.+b_.*Csch[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[x^m*(a+b*Csch[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x]
```

2: $\int (e\,x)^m \left(a+b\,\text{Sech}\left[c+d\,x^n\right]\right)^p dx$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} == 0$

Rule:

$$\int (e\,x)^m \left(a+b\,\text{Sech}\left[c+d\,x^n\right]\right)^p dx \longrightarrow \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a+b\,\text{Sech}\left[c+d\,x^n\right]\right)^p dx$$

Program code:

```
Int[(e_*x_)^m_.*(a_.+b_.*Sech[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Sech[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

```
Int[(e_*x_)^m_.*(a_.+b_.*Csch[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Csch[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

N: $\int (e\,x)^m\,(a+b\,\mathrm{Sech}[u])^p\,dx$ when $u==c+d\,x^n$

Derivation: Algebraic normalization

Rule: If $u==c+d\,x^n$, then

$$\int (e\,x)^m\,(a+b\,\mathrm{Sech}[u])^p\,dx \longrightarrow \int (e\,x)^m\,\left(a+b\,\mathrm{Sech}\left[c+d\,x^n\right]\right)^p\,dx$$

Program code:

```
Int[(e_*x_)^m_.*(a_.+b_.*Sech[u_])^p_.,x_Symbol] :=
  Int[(e*x)^m*(a+b*Sech[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

```
Int[(e_*x_)^m_.*(a_.+b_.*Csch[u_])^p_.,x_Symbol] :=
  Int[(e*x)^m*(a+b*Csch[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

Rules for integrands of the form $x^m \operatorname{Sech}[a+b x^n]^p \operatorname{Sinh}[a+b x^n]$

1: $\int x^m \operatorname{Sech}[a+b x^n]^p \operatorname{Sinh}[a+b x^n]\, dx$ when $n \in \mathbb{Z} \wedge m-n \geq 0 \wedge p \neq 1$

Derivation: Integration by parts

Rule: If $n \in \mathbb{Z} \wedge m-n \geq 0 \wedge p \neq 1$, then

$$\int x^m \operatorname{Sech}[a+b x^n]^p \operatorname{Sinh}[a+b x^n]\, dx \longrightarrow -\frac{x^{m-n+1} \operatorname{Sech}[a+b x^n]^{p-1}}{b n (p-1)} + \frac{m-n+1}{b n (p-1)} \int x^{m-n} \operatorname{Sech}[a+b x^n]^{p-1}\, dx$$

Program code:

```
Int[x_^m_.*Sech[a_.+b_.*x_^n_.]^p_*Sinh[a_.+b_.*x_^n_.],x_Symbol] :=
  -x^(m-n+1)*Sech[a+b*x^n]^(p-1)/(b*n*(p-1)) +
  (m-n+1)/(b*n*(p-1))*Int[x^(m-n)*Sech[a+b*x^n]^(p-1),x] /;
FreeQ[{a,b,p},x] && IntegerQ[n] && GeQ[m-n,0] && NeQ[p,1]
```

```
Int[x_^m_.*Csch[a_.+b_.*x_^n_.]^p_*Cosh[a_.+b_.*x_^n_.],x_Symbol] :=
  -x^(m-n+1)*Csch[a+b*x^n]^(p-1)/(b*n*(p-1)) +
  (m-n+1)/(b*n*(p-1))*Int[x^(m-n)*Csch[a+b*x^n]^(p-1),x] /;
FreeQ[{a,b,p},x] && IntegerQ[n] && GeQ[m-n,0] && NeQ[p,1]
```